# Maintainer: Antonio Rojas <arojas@archlinux.org>

pkgname=lager
pkgver=0.1.1
pkgrel=1
pkgdesc='C++ library for value-oriented design using the unidirectional data-flow architecture'
arch=(any)
url='https://sinusoid.es/lager'
license=(Boost)
makedepends=(cmake)
source=(https://github.com/arximboldi/$pkgname/archive/v$pkgver/$pkgname-$pkgver.tar.gz)
sha256sums=('9e4743c3fe2c95c1653c3fd088a2200108f09d758725697831852dc91d15d174')

build() {
  cmake -B build -S $pkgname-$pkgver \
    -DCMAKE_INSTALL_PREFIX=/usr \
    -Dlager_BUILD_EXAMPLES=OFF \
    -Dlager_BUILD_TESTS=OFF
  cmake --build build
}

package() {
  DESTDIR="$pkgdir" cmake --install build
}
